
0 To Begin With…
Prerequisites

If you have done the homework requested before
this summer school, you should have familiarised
yourself with

• values and types, and basic list processing,

• basics of type classes,

• defining functions by pattern matching,

• guards, case, local definitions by where and let,

• recursive definition of functions,

• and higher order functions.

Recommanded Textbooks

• Introduction to Functional Programming using
Haskell [Bir98]. My recommended book. Covers
equational reasoning very well.

• Programming in Haskell [Hut16]. A thin but com-
plete textbook.

• Learn You a Haskell for Great Good! [Lip11], a
nice tutorial with cute drawings!

• Real World Haskell [OSG98].

• Algorithm Design with Haskell [BG20].

1 Definition and Proof by Induc-
tion

Total Functional Programming

• The next few lectures concerns inductive defini-
tions and proofs of datatypes and programs.

• While Haskell provides allows one to define non-
terminating functions, infinite data structures, for
now we will only consider its total, finite frag-
ment.

• That is, we temporarily

– consider only finite data structures,

– demand that functions terminate for all
value in its input type, and

– provide guidelines to construct such func-
tions.

• Infinite datatypes and non-termination can be
modelled with more advanced theory, which we
cannot cover in this course.

1.1 Induction on Natural Numbers
Recalling “Mathematical Induction”

• Let P be a predicate on natural numbers.

– What is a predicate? Such a predicate can
be seen as a function of type Nat→ Bool.

– So far, we see Haskell functions as simple
mathematical functions too.

– However, Haskell functions will turn out to
be more complex than mere mathematical
functions later. To avoid confusion, we do
not use the notation Nat → Bool for predi-
cates.

• We’ve all learnt this principle of proof by induc-
tion: to prove that P holds for all natural num-
bers, it is sufficient to show that

– P 0 holds;

– P (1 + n) holds provided that P n does.

1.1.1 Proof by Induction

Proof by Induction on Natural Numbers

• We can see the above inductive principle as a re-
sult of seeing natural numbers as defined by the
datatype 1

data Nat = 0 | 1+ Nat .

• That is, any natural number is either 0, or 1+ n
where n is a natural number.

1Not a real Haskell definition.

• In this lecture, 1+ is written in bold font to em-
phasise that it is a data constructor (as opposed
to the function (+), to be defined later, applied to
a number 1).

A Proof Generator
Given P 0 and P n ⇒ P (1+ n), how does one

prove, for example, P 3?

P (1+ (1+ (1+ 0)))
⇐ { P (1+ n)⇐ P n }

P (1+ (1+ 0))
⇐ { P (1+ n)⇐ P n }

P (1+ 0)
⇐ { P (1+ n)⇐ P n }

P 0 .

Having done math. induction can be seen as having
designed a program that generates a proof — given any
n :: Nat we can generate a proof of P n in the manner
above.

1.1.2 Inductively Definition of Functions

Inductively Defined Functions

• Since the type Nat is defined by two cases, it is
natural to define functions on Nat following the
structure:

exp :: Nat → Nat → Nat
exp b 0 = 1
exp b (1+ n) = b× exp b n .

• Even addition can be defined inductively

(+) :: Nat → Nat → Nat
0 + n = n
(1+ m) + n = 1+ (m+ n) .

• Exercise: define (×)?

A Value Generator
Given the definition of exp, how does one compute

exp b 3?

exp b (1+ (1+ (1+ 0)))
= { definition of exp }

b× exp b (1+ (1+ 0))
= { definition of exp }

b× b× exp b (1+ 0)
= { definition of exp }

b× b× b× exp b 0
= { definition of exp }

b× b× b× 1 .

It is a program that generates a value, for any n ::
Nat . Compare with the proof of P above.

Moral: Proving is Programming
An inductive proof is a program that generates a

proof for any given natural number.
An inductive program follows the same structure of

an inductive proof.
Proving and programming are very similar activi-

ties.

Without the n+ k Pattern

• Unfortunately, newer versions of Haskell aban-
doned the “n + k pattern” used in the previous
slides:

exp :: Int → Int → Int
exp b 0 = 1
exp b n = b× exp b (n− 1) .

• Nat is defined to be Int in MiniPrelude.hs.
Without MiniPrelude.hs you should use Int .

• For the purpose of this course, the pattern 1 +
n reveals the correspondence between Nat and
lists, and matches our proof style. Thus we will
use it in the lecture.

• Remember to remove them in your code.

Proof by Induction

• To prove properties about Nat , we follow the
structure as well.

• E.g. to prove that exp b (m + n) = exp b m ×
exp b n.

• One possibility is to preform induction onm. That
is, prove P m for all m :: Nat , where P m ≡
(∀n :: exp b (m+ n) = exp b m× exp b n).

Case m := 0. For all n, we reason:

exp b (0 + n)
= { defn. of (+) }

exp b n
= { defn. of (×) }

1× exp b n
= { defn. of exp }

exp b 0× exp b n .

We have thus proved P 0.

Case m := 1+ m. For all n, we reason:

exp b ((1+ m) + n)
= { defn. of (+) }

exp b (1+ (m+ n))
= { defn. of exp }

b× exp b (m+ n)
= { induction }

b× (exp b m× exp b n)
= { (×) associative }

(b× exp b m)× exp b n
= { defn. of exp }

exp b (1+ m)× exp b n .

We have thus proved P (1+ m), given P m.

Structure Proofs by Programs

• The inductive proof could be carried out
smoothly, because both (+) and exp are defined
inductively on its lefthand argument (of type
Nat ).

• The structure of the proof follows the structure of
the program, which in turns follows the structure
of the datatype the program is defined on.

Lists and Natural Numbers

• We have yet to prove that (×) is associative.

• The proof is quite similar to the proof for associa-
tivity of (++), which we will talk about later.

• In fact, Nat and lists are closely related in struc-
ture.

• Most of us are used to think of numbers as atomic
and lists as structured data. Neither is necessarily
true.

• For the rest of the course we will demonstrate in-
duction using lists, while taking the properties for
Nat as given.

1.1.3 A Set-Theoretic Explanation of Induction

An Inductively Defined Set?

• For a set to be “inductively defined”, we usually
mean that it is the smallest fixed-point of some
function.

• What does that maen?

Fixed-Point and Prefixed-Point

• A fixed-point of a function f is a value x such that
f x = x.

• Theorem. f has fixed-point(s) if f is a monotonic
function defined on a complete lattice.

– In general, given f there may be more than
one fixed-point.

• A prefixed-point of f is a value x such that f x ⩽
x.

– Apparently, all fixed-points are also
prefixed-points.

• Theorem. the smallest prefixed-point is also the
smallest fixed-point.

Example: Nat

• Recall the usual definition: Nat is defined by the
following rules:

1. 0 is in Nat ;

2. if n is in Nat , so is 1+ n;

3. there is no other Nat .

• If we define a function F from sets to sets: F X =
{0}∪{1+ n | n ∈ X}, 1. and 2. above means that
F Nat ⊆ Nat . That is, Nat is a prefixed-point of
F .

• 3. means that we want the smallest such prefixed-
point.

• Thus Nat is also the least (smallest) fixed-point of
F .

Least Prefixed-Point
Formally, let F X = {0} ∪ {1+ n | n ∈ X}, Nat is

a set such that

F Nat ⊆ Nat , (1)

(∀X : F X ⊆ X ⇒ Nat ⊆ X) , (2)

where (1) says that Nat is a prefixed-point of F , and
(2) it is the least among all prefixed-points of F .

Mathematical Induction, Formally

• Given property P , we also denote by P the set of
elements that satisfy P .

• That P 0 and P n ⇒ P (1+n) is equivalent to
{0} ⊆ P and {1+ n | n ∈ P} ⊆ P ,

• which is equivalent to F P ⊆ P . That is, P is a
prefixed-point!

• By (2) we have Nat ⊆ P . That is, all Nat satisfy
P !

• This is “why mathematical induction is correct.”

Coinduction?
There is a dual technique called coinduction where,

instead of least prefixed-points, we talk about greatest
postfixed points. That is, largest x such that x ⩽ f x.

With such construction we can talk about infinite
data structures.

1.2 Induction on Lists
Inductively Defined Lists

• Recall that a (finite) list can be seen as a datatype
defined by: 2

data List a = [] | a : List a .

• Every list is built from the base case [ ], with ele-
ments added by (:) one by one: [1, 2, 3] = 1 : (2 :
(3 : [ ])).

All Lists Today are Finite
But what about infinite lists?

• For now let’s consider finite lists only, as having
infinite lists make the semantics much more com-
plicated. 3

• In fact, all functions we talk about today are total
functions. No ⊥ involved.

Set-Theoretically Speaking…
The type List a is the smallest set such that

1. [ ] is in List a;

2. if xs is in List a and x is in a, x : xs is in List a
as well.

Inductively Defined Functions on Lists

• Many functions on lists can be defined according
to how a list is defined:

sum :: List Int → Int
sum [ ] = 0
sum (x : xs) = x+ sum xs .

map :: (a→ b)→ List a→ List b
map f [ ] = [ ]
map f (x : xs) = F X : map f xs .

– sum [1..10] = 55

– map (1+) [1, 2, 3, 4] = [2, 3, 4, 5]
2Not a real Haskell definition.
3What does that mean? Other courses in FLOLAC might cover

semantics in more detail.

1.2.1 Append, and Some of Its Properties

List Append

• The function (++) appends two lists into one

(++) :: List a→ List a→ List a
[ ] ++ ys = ys
(x : xs)++ ys = x : (xs ++ ys) .

• Compare the definition with that of (+)!

Proof by Structural Induction on Lists

• Recall that every finite list is built from the base
case [ ], with elements added by (:) one by one.

• To prove that some property P holds for all finite
lists, we show that

1. P [ ] holds;
2. forall x and xs , P (x : xs) holds provided

that P xs holds.

For a Particular List…
Given P [ ] and P xs ⇒ P (x : xs), for all x and xs ,

how does one prove, for example, P [1, 2, 3]?

P (1 : 2 : 3 : [ ])
⇐ { P (x : xs)⇐ P xs }

P (2 : 3 : [ ])
⇐ { P (x : xs)⇐ P xs }

P (3 : [ ])
⇐ { P (x : xs)⇐ P xs }

P [ ] .

Appending is Associative
To prove that xs ++(ys ++ zs) = (xs ++ ys)++ zs .
Let P xs = (∀ys, zs :: xs ++(ys ++ zs) =

(xs ++ ys)++ zs), we prove P by induction on xs .
Case xs := [ ]. For all ys and zs, we reason:

[ ] ++(ys ++ zs)
= { defn. of (++) }

ys ++ zs
= { defn. of (++) }

([ ] ++ ys)++ zs .

We have thus proved P [].
Case xs := x : xs . For all ys and zs, we reason:

(x : xs)++(ys ++ zs)
= { defn. of (++) }

x : (xs ++(ys ++ zs))
= { induction }

x : ((xs ++ ys)++ zs)
= { defn. of (++) }

(x : (xs ++ ys))++ zs
= { defn. of (++) }

((x : xs)++ ys)++ zs .

We have thus proved P (x : xs), given P xs.

DoWe Have To Be So Formal?

• In our style of proof, every step is given a reason.
Do we need to be so pedantic?

• Being formal helps you to do the proof:

– In the proof of exp b (m + n) = exp b m ×
exp b n, we expect that we will use induction
to somewhere. Therefore the first part of the
proof is to generate exp b (m+ n).

– In the proof of associativity, we were work-
ing toward generating xs ++(ys ++ zs).

• By being formal we can work on the form, not the
meaning. Like how we solved the knight/knave
problem

• Being formal actually makes the proof easier!

• Make the symbols do the work.

Length

• The function length defined inductively:

length :: List a→ Nat
length [ ] = 0
length (x : xs) = 1+ (length xs) .

• Exercise: prove that length distributes into (++):

length (xs ++ ys) = length xs + length ys

Concatenation

• While (++) repeatedly applies (:), the function
concat repeatedly calls (++):

concat :: List (List a)→ List a
concat [ ] = [ ]
concat (xs : xss) = xs ++ concat xss .

• Compare with sum .

• Exercise: prove sum · concat = sum ·map sum .

1.2.2 More Inductively Defined Functions

Definition by Induction/Recursion

• Rather than giving commands, in functional pro-
gramming we specify values; instead of perform-
ing repeated actions, we define values on induc-
tively defined structures.

• Thus induction (or in general, recursion) is the
only “control structure” we have. (We do identify
and abstract over plenty of patterns of recursion,
though.)

• Note Terminology: an inductive definition, as
we have seen, define “bigger” things in terms of
“smaller” things. Recursion, on the other hand, is
a more general term, meaning “to define one en-
tity in terms of itself.”

• To inductively define a function f on lists, we
specify a value for the base case (f [ ]) and, assum-
ing that f xs has been computed, consider how to
construct f (x : xs) out of f xs .

Filter

• filter p xs keeps only those elements in xs that
satisfy p.

filter :: (a→ Bool)→ List a→ List a
filter p [ ] = [ ]
filter p (x : xs) | p x = x : filter p xs

| otherwise = filter p xs .

Take and Drop

• Recall take and drop, which we used in the previ-
ous exercise.

take :: Nat → List a→ List a
take 0 xs = [ ]
take (1+ n) [ ] = [ ]
take (1+ n) (x : xs) = x : take n xs .

drop :: Nat → List a→ List a
drop 0 xs = xs
drop (1+ n) [ ] = [ ]
drop (1+ n) (x : xs) = drop n xs .

• Prove: take n xs ++ drop n xs = xs , for all n and
xs .

TakeWhile and DropWhile

• takeWhile p xs yields the longest prefix of xs such
that p holds for each element.

takeWhile :: (a→ Bool)→ List a→ List a
takeWhile p [ ] = [ ]
takeWhile p (x : xs) | p x = x : takeWhile p xs

| otherwise = [ ] .

• dropWhile p xs drops the prefix from xs .

dropWhile :: (a→ Bool)→ List a→ List a
dropWhile p [ ] = [ ]
dropWhile p (x : xs) | p x = dropWhile p xs

| otherwise = x : xs .

• Prove: takeWhile p xs ++ dropWhile p xs = xs .

List Reversal

• reverse [1, 2, 3, 4] = [4, 3, 2, 1].

reverse :: List a→ List a
reverse [ ] = [ ]
reverse (x : xs) = reverse xs ++[x] .

All Prefixes and Suffixes

• inits [1, 2, 3] = [[ ], [1], [1, 2], [1, 2, 3]]

inits :: List a→ List (List a)
inits [ ] = [[ ]]
inits (x : xs) = [ ] : map (x :) (inits xs) .

• tails [1, 2, 3] = [[1, 2, 3], [2, 3], [3], [ ]]

tails :: List a→ List (List a)
tails [ ] = [[ ]]
tails (x : xs) = (x : xs) : tails xs .

Totality

• Structure of our definitions so far:

f [ ] = . . .
f (x : xs) = . . . f xs . . .

– Both the empty and the non-empty cases
are covered, guaranteeing there is a match-
ing clause for all inputs.

– The recursive call is made on a “smaller” ar-
gument, guranteeing termination.

• Together they guarantee that every input is
mapped to some output. Thus they define total
functions on lists.

1.2.3 Other Patterns of Induction

Variations with the Base Case

• Some functions discriminate between several
base cases. E.g.

fib :: Nat → Nat
fib 0 = 0
fib 1 = 1
fib (2 + n) = fib (1+n) + fib n .

• Some functions make more sense when it is de-
fined only on non-empty lists:

f [x] = . . .
f (x : xs) = . . .

• What about totality?

– They are in fact functions defined on a dif-
ferent datatype:

data List+ a = Singleton a | a : List+ a .

– We do not want to define map, filter again
for List+ a. Thus we reuse List a and pre-
tend that we were talking about List+ a.

– It’s the same with Nat . We embedded Nat
into Int .

– Ideally we’d like to have some form of sub-
typing. But that makes the type system
more complex.

Lexicographic Induction

• It also occurs often that we perform lexicographic
induction on multiple arguments: some argu-
ments decrease in size, while others stay the
same.

• E.g. the function merge merges two sorted lists
into one sorted list:

merge :: List Int → List Int → List Int
merge [ ] [ ] = [ ]
merge [ ] (y : ys) = y : ys
merge (x : xs) [ ] = x : xs
merge (x : xs) (y : ys) | x ⩽ y = x : merge xs (y : ys)

| otherwise = y : merge (x : xs) ys .

Zip
Another example:

zip :: List a→ List b→ List (a, b)
zip [ ] [ ] = [ ]
zip [ ] (y : ys) = [ ]
zip (x : xs) [ ] = [ ]
zip (x : xs) (y : ys) = (x, y) : zip xs ys .

Non-Structural Induction

• In most of the programs we’ve seen so far, the re-
cursive call are made on direct sub-components
of the input (e.g. f (x : xs) = ..f xs..). This is
called structural induction.

– It is relatively easy for compilers to recog-
nise structural induction and determine that
a program terminates.

• In fact, we can be sure that a program terminates
if the arguments get “smaller” under some (well-
founded) ordering.

Mergesort

• In the implemenation of mergesort below, for ex-
ample, the arguments always get smaller in size.

msort :: List Int → List Int
msort [ ] = [ ]
msort [x] = [x]
msort xs = merge (msort ys) (msort zs) ,
where n = length xs ‘div ‘ 2

ys = take n xs
zs = drop n xs .

– What if we omit the case for [x]?

• If all cases are covered, and all recursive calls are
applied to smaller arguments, the program de-
fines a total function.

A Non-Terminating Definition

• Example of a function, where the argument to the
recursive does not reduce in size:

f :: Int → Int
f 0 = 0
f n = f n .

• Certainly f is not a total function. Do such defini-
tions “mean” something? We will talk about these
later.

1.3 User Defined Inductive Datatypes
Internally Labelled Binary Trees

• This is a possible definition of internally labelled
binary trees:

data ITree a = Null | Node a (ITree a) (ITree a) ,

• on which we may inductively define functions:

sumT :: ITree Nat → Nat
sumT Null = 0
sumT (Node x t u) = x+ sumT t+ sumT u .

Exercise: given (↓) :: Nat → Nat → Nat , which
yields the smaller one of its arguments, define the fol-
lowing functions

1. minT :: Tree Nat → Nat , which computes the
minimal element in a tree.

2. mapT :: (a → b) → Tree a → Tree b, which
applies the functional argument to each element
in a tree.

3. Can you define (↓) inductively on Nat? 4

4In the standard Haskell library, (↓) is called min .

Induction Principle for Tree

• What is the induction principle for Tree?

• To prove that a predicate P on Tree holds for ev-
ery tree, it is sufficient to show that

1. P Null holds, and;

2. for every x, t, and u, if P t and P u holds,
P (Node x t u) holds.

• Exercise: prove that for all n and t,
minT (mapT (n+) t) = n + minT t. That is,
minT ·mapT (n+) = (n+) ·minT .

Induction Principle for Other Types

• Recall that data Bool = False | True . Do we
have an induction principle for Bool?

• To prove a predicate P on Bool holds for all
booleans, it is sufficient to show that

1. P False holds, and

2. P True holds.

• Well, of course.

• What about (A×B)? How to prove that a pred-
icate P on (A×B) is always true?

• One may prove some property P1 on A and some
property P2 on B, which together imply P .

• That does not say much. But the “induction prin-
ciple” for products allows us to extract, from a
proof of P , the proofs P1 and P2.

• Every inductively defined datatype comes with its
induction principle.

• We will come back to this point later.

2 Program Derivation

2.1 Some Comments on Efficiency
Data Representation

• So far we have (surprisingly) been talking about
mathematics without much concern regarding ef-
ficiency. Time for a change.

• Take lists for example. Recall the definition:
data List a = [ ] | a : List a.

• Our representation of lists is biased. The left most
element can be fetched immediately.

– Thus. (:), head , and tail are constant-time
operations, while init and last takes linear-
time.

• In most implementations, the list is represented
as a linked-list.

List Concatenation Takes Linear Time

• Recall (++):

[ ] ++ ys = ys
(x : xs)++ ys = x : (xs ++ ys)

• Consider [1, 2, 3]++[4, 5]:

(1 : 2 : 3 : [ ])++(4 : 5 : [ ])
= 1 : ((2 : 3 : [ ])++(4 : 5 : [ ]))
= 1 : 2 : ((3 : [ ])++(4 : 5 : [ ]))
= 1 : 2 : 3 : ([ ] ++(4 : 5 : [ ]))
= 1 : 2 : 3 : 4 : 5 : [ ]

• (++) runs in time proportional to the length of its
left argument.

Full Persistency

• Compound data structures, like simple values, are
just values, and thus must be fully persistent.

• That is, in the following code:

let xs = [1, 2, 3]
ys = [4, 5]
zs = xs ++ ys

in . . . body . . .

• The body may have access to all three values.
Thus ++ cannot perform a destructive update.

Linked v.s. Block Data Structures

• Trees are usually represented in a similar manner,
through links.

• Fully persistency is easier to achieve for such
linked data structures.

• Accessing arbitrary elements, however, usually
takes linear time.

• In imperative languages, constant-time random
access is usually achieved by allocating lists (usu-
ally called arrays in this case) in a consecutive
block of memory.

• Consider the following code, where xs is an array
(implemented as a block), and ys is like xs , apart
from its 10th element:

let xs = [1..100]
ys = update xs 10 20

in . . . body . . .

• To allow access to both xs and ys in body , the
update operation has to duplicate the entire ar-
ray.

• Thus people have invented some smart data
structure to do so, in around O(log n) time.

• On the other hand, update may simply overwrite
xs if we can somehow make sure that nobody
other than ys uses xs .

• Both are advanced topics, however.

Another Linear-Time Operation

• Taking all but the last element of a list:

init [x] = [ ]
init (x : xs) = x : init xs

• Consider init [1, 2, 3, 4]:

init (1 : 2 : 3 : 4 : [ ])
= 1 : init (2 : 3 : 4 : [ ])
= 1 : 2 : init (3 : 4 : [ ])
= 1 : 2 : 3 : init (4 : [ ])
= 1 : 2 : 3 : [ ]

Sum, Map, etc

• Functions like sum , maximum , etc. needs to tra-
verse through the list once to produce a result. So
their running time is definitely O(n), where n is
the length of the list.

• If f takes time O(t), map f takes time O(n × t)
to complete. Similarly with filter p.

– In a lazy setting, map f produces its first re-
sult in O(t) time. We won’t need lazy fea-
tures for now, however.

2.2 Expand/Reduce Transformation
Sum of Squares

• Given a sequence a1,a2,. . . ,an, compute a21 +
a22 + . . . + a2n. Specification: sumsq = sum ·
map square.

• The spec. builds an intermediate list. Can we
eliminate it?

Figure 1: How (++) allocates new (:) cells in the heap.

• The input is either empty or not. When it is
empty:

sumsq [ ]
= { definition of sumsq }

(sum ·map square) [ ]
= { function composition }

sum (map square [ ])
= { definition of map }

sum [ ]
= { definition of sum }

0

Sum of Squares, the Inductive Case

• Consider the case when the input is not empty:

sumsq (x : xs)
= { definition of sumsq }

sum (map square (x : xs))
= { definition of map }

sum (square x : map square xs)
= { definition of sum }

square x+ sum (map square xs)
= { definition of sumsq }

square x+ sumsq xs

Alternative Definition for sumsq

• From sumsq = sum · map square, we have
proved that

sumsq [ ] = 0
sumsq (x : xs) = square x+ sumsq xs

• Equivalently, we have shown that
sum ·map square is a solution of

f [ ] = 0
f (x : xs) = square x+ f xs

• However, the solution of the equations above is
unique.

• Thus we can take it as another definition of
sumsq. Denotationally it is the same function;
operationally, it is (slightly) quicker.

• Exercise: try calculating an inductive definition of
count .

Remark: Why Functional Programming?

• Time to muse on the merits of functional pro-
gramming. Why functional programming?

– Algebraic datatype? List comprehension?
Lazy evaluation? Garbage collection? These
are just language features that can be mi-
grated.

– No side effects.5 But why taking away a lan-
guage feature?

• By being pure, we have a simpler semantics in
which we are allowed to construct and reason
about programs.

– In an imperative language we do not even
have f 4 + f 4 = 2× f 4.

• Ease of reasoning. That’s the main benefit we get.

Example: Computing Polynomial
Given a list as = [a0, a1, a2 . . . an] and x :: Int, the

aim is to compute:

a0 + a1x+ a2x
2 + ...+ anx

n.

This can be specified by

poly x as = sum (zipWith (×) as (iterate (×x ) 1)) ,

where iterate can be defined by

iterate :: (a → a)→ a → List a
iterate f x = x :map f (iterate f x ) .

Iterating a List
To get some intuition about iterate let us try ex-

panding it:

5Unless introduced in disciplined ways. For example, through a
monad.

iterate f x
= { definition of iterate }
x :map f (iterate f x )

= { definition of map }
x :map f (x :map f (iterate f x ))

= { map fusion }
x : f x :map (f · f ) (iterate f x )

= { definitions of iterate and map }
x : f x : f (f x ) :map (f · f ) (map f (iterate f x ))

= { map fusion }
x : f x : f (f x ) :map (f · f · f ) (iterate f x ) . . .

Zipping with a Binary Operator
While iterate generate a list, it is immediately trun-

cated by zipWith :

zipWith :: (a → b → c)→
List a → List b → List c

zipWith (⊕) [ ] = [ ]
zipWith (⊕) (x : xs) [ ] = [ ]
zipWith (⊕) (x : xs) (y : ys) =

x ⊕ y : zipWith (⊕) xs ys .

Running the Specification
Try expanding poly x [a, b, c, d ], we get

poly x [a, b, c, d ]
= sum (zipWith (×) [a, b, c, d ] (iterate (×x ) 1))
= { expanding iterate }

sum (zipWith (×) [a, b, c, d ]
(1 : (1× x ) : (1× x × x ) : (1× x × x × x ) :
map (×x )4 (iterate (×x ) 1)))

= a + b × x + c × x × x + d × x × x × x .

where f 4 denotes f · f · f · f .
As the list gets longer, we get more (×x ) accumu-

lating. Can we do better?

The main calculation
poly x (a : as)

= { definition of poly }
sum (zipWith (×) (a : as) (iterate (×x ) 1))

= { definition of iterate }
sum (zipWith (×) (a : as)
(1 :map (×x ) (iterate (×x ) 1)))

= { definitions of zipWith and sum }
a + sum (zipWith (×) as
(map (×x ) (iterate (×x ) 1)))

= { see below }
a + sum (map (×x ) (zipWith (×)
as (iterate (×x ) 1)))

= { sum ·map (×x ) = (×x ) · sum }
a + (sum (zipWith (×) as (iterate (×x ) 1)))× x

= { definition of poly }
a + (poly x as)× x .

Zip-Map Exchange
In the 4th step we used the property

zipWith (×) as · map (×x ) = map (×x ) ·
zipWith (×) as .

It applies to any operator (⊗) that is associative. For
an intuitive understanding:

zipWith (⊗) [a, b, c ] (map (⊗x ) [d , e, f ])
= [a ⊗ (d ⊗ x ), b ⊗ (e ⊗ x ), c ⊗ (f ⊗ x )]
= { associativity: m ⊗ (n ⊗ k) = (m ⊗ n)⊗ k }

[(a ⊗ d)⊗ x , (b ⊗ e)⊗ x , (c ⊗ f )⊗ x ]
= map (⊗x ) (zipWith (⊗) [a, b, c ] [d , e, f ]) .

We can do a formal proof if we want.

Distributivity
In the 5th step we used the property sum ·

map (×x ) = (×x ) · sum . For that we need distribu-
tivity between addition and multiplication.

We used that law to push sum to the right.
This is the crucial property that allows us to speed

up poly : we are allowed to factor out common (×x ).

Computing Polynomial
To conclude, we get:

poly x [ ] = 0
poly x (a : as) = a + (poly as)× x ,

which uses a linear number of (×).

Let the Symbols Do the Work!
How do we know what laws to use or to assume?
By observing the form of the expressions. Let the

symbols do the work.

2.3 Tupling

Steep Lists

• A steep list is a list in which every element is larger
than the sum of those to its right:

steep :: List Int → Bool
steep [ ] = True
steep (x : xs) = steep xs ∧ x > sum xs .

• The definition above, if executed directly, is an
O(n2) program. Can we do better?

• Just now we learned to construct a generalised
function which takes more input. This time, we
try the dual technique: to construct a function re-
turning more results.

Generalise by Returning More

• Recall that fst (a, b) = a and snd (a, b) = b.

• It is hard to quickly compute steep alone. But if
we define

steepsum :: List Int → (Bool × Int)
steepsum xs = (steep xs, sum xs),

• and manage to synthesise a quick definition of
steepsum , we can implement steep by steep =
fst · steepsum .

• We again proceed by case analysis. Trivially,

steepsum [ ] = (True, 0).

Deriving for the Non-Empty Case
For the case for non-empty inputs:

steepsum (x : xs)
= { definition of steepsum }

(steep (x : xs), sum (x : xs))
= { definitions of steep and sum }

(steep xs ∧ x > sum xs, x+ sum xs)
= { extracting sub-expressions involving xs }

let (b, y) = (steep xs, sum xs)
in (b ∧ x > y, x+ y)

= { definition of steepsum }
let (b, y) = steepsum xs
in (b ∧ x > y, x+ y).

Synthesised Program
We have thus come up with a O(n) time program:

steep = fst · steepsum
steepsum [ ] = (True, 0)
steepsum (x : xs) = let (b, y) = steepsum xs

in (b ∧ x > y, x+ y),

BeingQuicker by Doing More?

• A more generalised program can be implemented
more efficiently?

– A common phenomena! Sometimes the less
general function cannot be implemented in-
ductively at all!

– It also often happens that a theorem needs
to be generalised to be proved. We will see
that later.

• An obvious question: how do we know what gen-
eralisation to pick?

• There is no easy answer — finding the right gen-
eralisation one of the most difficulty act in pro-
gramming!

• Sometimes we simply generalise by examining
the form of the formula.

2.4 Accumulating Parameters

Reversing a List

• The function reverse is defined by:

reverse [ ] = [ ],
reverse (x : xs) = reverse xs ++[x].

• E.g. reverse [1, 2, 3, 4] =
((([ ] ++[4])++[3])++[2])++[1] = [4, 3, 2, 1].

• But how about its time complexity? Since (++) is
O(n), it takes O(n2) time to revert a list this way.

• Can we make it faster?

2.4.1 Fast List Reversal

Introducing an Accumulating Parameter

• Let us consider a generalisation of reverse . De-
fine:

revcat :: List a→ List a→ List a
revcat xs ys = reverse xs ++ ys .

• If we can construct a fast implementation of
revcat , we can implement reverse by:

reverse xs = revcat xs [ ].

Reversing a List, Base Case
Let us use our old trick. Consider the case when xs

is [ ]:

revcat [ ] ys
= { definition of revcat }

reverse [ ] ++ ys
= { definition of reverse }

[ ] ++ ys
= { definition of (++) }

ys.

Reversing a List, Inductive Case
Case x : xs :

revcat (x : xs) ys
= { definition of revcat }

reverse (x : xs)++ ys
= { definition of reverse }

(reverse xs ++[x]) ++ ys
= { since (xs ++ ys)++ zs = xs ++(ys ++ zs) }

reverse xs ++([x] ++ ys)
= { definition of revcat }

revcat xs (x : ys).

Linear-Time List Reversal

• We have therefore constructed an implementa-
tion of revcat which runs in linear time!

revcat [ ] ys = ys
revcat (x : xs) ys = revcat xs (x : ys).

• A generalisation of reverse is easier to implement
than reverse itself? How come?

• If you try to understand revcat operationally, it is
not difficult to see how it works.

– The partially reverted list is accumulated in
ys.

– The initial value of ys is set by reverse xs =
revcat xs [ ].

– Hmm… it is like a loop, isn’t it?

2.4.2 Tail Recursion and Loops

Tracing Reverse

reverse [1, 2, 3, 4]
= revcat [1, 2, 3, 4] [ ]
= revcat [2, 3, 4] [1]
= revcat [3, 4] [2, 1]
= revcat [4] [3, 2, 1]
= revcat [ ] [4, 3, 2, 1]
= [4, 3, 2, 1]

reverse xs = revcat xs [ ]
revcat [ ] ys = ys
revcat (x : xs) ys = revcat xs (x : ys)

xs, ys ← XS , [ ];
while xs ̸= [ ] do

xs, ys ← (tail xs), (head xs : ys);
return ys

Tail Recursion

• Tail recursion: a special case of recursion in which
the last operation is the recursive call.

f x1 . . . xn = {base case}
f x1 . . . xn = f x′

1 . . . x′
n

• To implement general recursion, we need to keep
a stack of return addresses. For tail recursion, we
do not need such a stack.

• Tail recursive definitions are like loops. Each xi is
updated to x′

i in the next iteration of the loop.

• The first call to f sets up the initial values of each
xi.

Accumulating Parameters

• To efficiently perform a computation (e.g.
reverse xs), we introduce a generalisation with
an extra parameter, e.g.:

revcat xs ys = reverse xs ++ ys .

• Try to derive an efficient implementation of the
generalised function. The extra parameter is usu-
ally used to “accumulate” some results, hence the
name.

– To make the accumulation work, we usually
need some kind of associativity.

• A technique useful for, but not limited to, con-
structing tail-recursive definition of functions.

Accumulating Parameter: Another Example

• Recall the “sum of squares” problem:

sumsq [ ] = 0
sumsq (x : xs) = square x+ sumsq xs .

• The program still takes linear space (for the stack
of return addresses). Let us construct a tail recur-
sive auxiliary function.

• Introduce ssp xs n = sumsq xs + n.

• Initialisation: sumsq xs = ssp xs 0.

• Construct ssp:

ssp [ ] n = 0 + n = n
ssp (x : xs) n = (square x+ sumsq xs) + n

= sumsq xs + (square x+ n)
= ssp xs (square x+ n).

2.5 Conclusions
Conclusions

• Let the symbols do the work!

– Algebraic manipulation helps us to sepa-
rate the more mechanical parts of reasoning,
from the parts that needs real innovation.

• For more examples of fun program calculation,
see Bird [Bir10].

• For a more systematic study of algorithms using
functional program reasoning, see Bird and Gib-
bons [BG20].

3 Folds On Lists
A Common Pattern We’ve Seen Many Times. . .

sum [ ] = 0
sum (x : xs) = x+ sum xs

length [ ] = 0
length (x : xs) = 1 + length xs

map f [ ] = [ ]
map f (x : xs) = f x : map f xs

This pattern is extracted and called foldr :

foldr f e [ ] = e,
foldr f e (x : xs) = f x (foldr f e xs).

3.1 The Ubiquitous foldr
Replacing Constructors

foldr f e [ ] = e
foldr f e (x : xs) = f x (foldr f e xs)

• One way to look at foldr (⊕) e is that it replaces
[ ] with e and (:) with (⊕):

foldr (⊕) e [1, 2, 3, 4]
= foldr (⊕) e (1 : (2 : (3 : (4 : [ ]))))

= 1⊕ (2⊕ (3⊕ (4⊕ e))).

• sum = foldr (+) 0.

• length = foldr (λx n.1 + n) 0.

• map f = foldr (λx xs .f x : xs) [ ].

• One can see that id = foldr (:) [ ].

Some Trivial Folds on Lists

• Function max returns the least upper bound of
elements in a list:

max [ ] = -∞,
max (x : xs) = x ↑ max xs .

max = foldr (↑) -∞.

• This function is actually called maximum in
the standard Haskell Prelude, while max returns
the maximum between its two arguments. For
brevity, we denote the former by max and the lat-
ter by (↑).

• Function prod returns the product of a list:

prod [ ] = 1,
prod (x : xs) = x× prod xs .

prod = foldr (×) 1.

• Function and returns the conjunction of a list:

and [ ] = true,
and (x : xs) = x ∧ and xs .

and = foldr (∧) true.

• Lets emphasise again that id on lists is a fold:

id [ ] = [ ],
id (x : xs) = x : id xs .

id = foldr (:) [ ].

Some Functions We Have Seen. . .

• (++ ys) = foldr (:) ys .

(++) :: [a]→ [a]→ [a]
[ ] ++ ys = ys
(x : xs)++ ys = x : (xs ++ ys) .

• concat = foldr (++) [ ].

concat :: [[a]]→ [a]
concat [ ] = [ ]
concat (xs : xss) = xs ++ concat xss .

Replacing Constructors

• Understanding foldr from its type. Recall

data [a] = [ ] | a : [a] .

• Types of the two constructors: [ ] :: [a], and (:) ::
a→ [a]→ [a].

• foldr replaces the constructors:

foldr :: (a→ b→ b)→ b→ [a]→ b
foldr f e [ ] = e
foldr f e (x : xs) = f x (foldr f e xs) .

3.2 The Fold-Fusion Theorem
Why Folds?

• “What are the three most important factors in
a programming language?” Abstraction, abstrac-
tion, and abstraction!

• Control abstraction, procedure abstraction, data
abstraction,. . . can programming patterns be ab-
stracted too?

• Program structure becomes an entity we can talk
about, reason about, and reuse.

– We can describe algorithms in terms of fold,
unfold, and other recognised patterns.

– We can prove properties about folds,

– and apply the proved theorems to all pro-
grams that are folds, either for compiler op-
timisation, or for mathematical reasoning.

• Among the theorems about folds, the most impor-
tant is probably the fold-fusion theorem.

The Fold-Fusion Theorem
The theorem is about when the composition of a

function and a fold can be expressed as a fold.

Theorem 1 (foldr -Fusion). Given f :: a → b → b,
e :: b, h :: b→ c, and g :: a→ c→ c, we have:

h · foldr f e = foldr g (h e) ,

if h (f x y) = g x (h y) for all x and y.

For program derivation, we are usually given h, f ,
and e, from which we have to construct g.

Tracing an Example
Let us try to get an intuitive understand of the the-

orem:

h (foldr f e [a, b, c])

= { definition of foldr }
h (f a (f b (f c e)))

= { since h (f x y) = g x (h y) }
g a (h (f b (f c e)))

= { since h (f x y) = g x (h y) }
g a (g b (h (f c e)))

= { since h (f x y) = g x (h y) }
g a (g b (g c (h e)))

= { definition of foldr }
foldr g (h e) [a, b, c] .

Sum of Squares, Again

• Consider sum · map square again. This time we
use the fact that map f = foldr (mf f) [ ], where
mf f x xs = f x : xs .

• sum · map square is a fold, if we can find
a ssq such that sum (mf square x xs) =
ssq x (sum xs). Let us try:

sum (mf square x xs)

= { definition of mf }
sum (square x : xs)

= { definition of sum }
square x+ sum xs

= { let ssq x y = square x+ y }
ssq x (sum xs) .

Therefore, sum ·map square = foldr ssq 0.

Sum of Squares, without Folds
Recall that this is how we derived the inductive case

of sumsq yesterday:

sumsq (x : xs)

= { definition of sumsq }
sum (map square (x : xs))

= { definition of map }
sum (square x : map square xs)

= { definition of sum }
square x+ sum (map square xs)

= { definition of sumsq }
square x+ sumsq xs .

Comparing the two derivations, by using fold-fusion
we supply only the “important” part.

More on Folds and Fold-fusion

• Compare the proof with the one yesterday. They
are essentially the same proof.

• Fold-fusion theorem abstracts away the common
parts in this kind of inductive proofs, so that we
need to supply only the “important” parts.

• Tupling can be seen as a kind of fold-fusion. The
derivation of steepsum, for example, can be seen
as fusing:

steepsum · id = steepsum · foldr (:) [ ].

– Recall that steepsum xs =
(steep xs, sum xs). Reformulating
steepsum into a fold allows us to com-
pute it in one traversal.

• Not every function can be expressed as a fold. For
example, tail :: [a]→ [a] is not a fold!

3.3 More Useful Functions Defined as
Folds

Longest Prefix

• The function call takeWhile p xs returns the
longest prefix of xs that satisfies p:

takeWhile p [ ] = [ ]
takeWhile p (x : xs) =

if p x then x : takeWhile p xs
else [ ] .

• E.g. takeWhile (⩽ 3) [1, 2, 3, 4, 5] = [1, 2, 3].

• It can be defined by a fold:

takeWhile p = foldr (tke p) [ ],
tke p x xs = if p x then x : xs else [ ].

• Its dual, dropWhile (⩽ 3) [1, 2, 3, 4, 5] = [4, 5], is
not a fold.

All Prefixes

• The function inits returns the list of all prefixes
of the input list:

inits [ ] = [[ ]],
inits (x : xs) = [ ] : map (x :) (inits xs).

• E.g. inits [1, 2, 3] = [[ ], [1], [1, 2], [1, 2, 3]].

• It can be defined by a fold:

inits = foldr ini [[ ]],
ini x xss = [ ] : map (x :) xss .

All Suffixes

• The function tails returns the list of all suffixes of
the input list:

tails [ ] = [[ ]],
tails (x : xs) = let (ys : yss) = tails xs

in (x : ys) : ys : yss .

• E.g. tails [1, 2, 3] = [[1, 2, 3], [2, 3], [3], [ ]].

• It can be defined by a fold:

tails = foldr til [[ ]],
til x (ys : yss) = (x : ys) : ys : yss .

Scan

• scanr f e = map (foldr f e) · tails .

• E.g.

scanr (+) 0 [1, 2, 3]

= map sum (tails [1, 2, 3])

= map sum [[1, 2, 3], [2, 3], [3], [ ]]

= [6, 5, 3, 0].

• Of course, it is slow to actually perform
map (foldr f e) separately. By fold-fusion, we
get a faster implementation:

scanr f e = foldr (sc f) [e],
sc f x (y : ys) = f x y : y : ys .

4 Folds on Other Algebraic
Datatypes

• Folds are a specialised form of induction.

• Inductive datatypes: types on which you can per-
form induction.

• Every inductive datatype give rise to its fold.

• In fact, an inductive type can be defined by its
fold.

Fold on Natural Numbers

• Recall the definition:

data Nat = 0 | 1+ Nat .

• Constructors: 0 :: Nat , (1+) :: Nat → Nat .

• What is the fold on Nat?

foldN :: (a→ a)→ a→ Nat → a
foldN f e 0 = e
foldN f e (1+ n) = f (foldN f e n) .

Examples of foldN

• (+n) = foldN (1+) n.

0 + n = n
(1+ m) + n = 1+ (m+ n) .

• (×n) = foldN (n+) 0.

0× n = 0
(1+ m)× n = n+ (m× n) .

• even = foldN not True .

even 0 = True
even (1+ n) = not (even n) .

Fold-Fusion for Natural Numbers

Theorem 2 (foldN -Fusion). Given f :: a→ a, e :: a,
h :: a→ b, and g :: b→ b, we have:

h · foldN f e = foldN g (h e) ,

if h (f x) = g (h x) for all x.

Exercise: fuse even into (+)?

Folds on Trees

• Example: internally labelled binary tree:

data ITree a = Null
| Node a (ITree a) (ITree a) .

• Fold for ITree:

foldIT :: (a → b → b → b)→ b → ITree a → b
foldIT f e Null = e
foldIT f e (Node a t u) =
f a (foldIT f e t) (foldIT f e u) .

Folds on Trees

• Example: externally labelled binary tree:

• Some datatypes for trees:

data ETree a = Tip a
| Bin (ETree a) (ETree a) .

• Fold for ETree:

foldET :: (b → b → b)→ (a → b)
→ ETree a → b

foldET f g (Tip x ) = g x
foldET f g (Bin t u) =
f (foldET f g t) (foldET f g u) .

Some Simple Functions on Trees

• To compute the size of an ITree:

sizeIT = foldIT (λx m n → 1+ (m + n)) 0 .

• To sum up labels in an ETree:

sizeET = foldET (+) id .

• To compute a list of all labels in an ITree and an
ETree:

flattenIT =
foldIT (λx xs ys → xs ++ [x ] ++ ys) [ ] ,

flattenET = foldET (++) (λx → [x ]) .

• Exercise: what are the fusion theorems for
foldIT and foldET ?

5 Maximum Segment Sum
• The maximum segment sum is a classical problem,

often used to demonstrate the effectness of pro-
gram derivation.

• Given: a list of numbers — positive, zero, or nega-
tive.

• Compute: the maximum possible sum of a con-
secutive segment of the list.

Specifying Maximum Segment Sum

• A segment can be seen as a prefix of a suffix.

• The function segs computes the list of all the seg-
ments.

segs = concat ·map inits · tails .

• Therefore, mss is specified by:

mss = max ·map sum · segs .

The Derivation!
We reason:

max ·map sum · concat ·map inits · tails
= { since map f · concat = concat ·map (map f ) }
map · concat ·map (map sum) ·
map inits · tails

= { since max · concat = max ·map max }
max ·map max ·map (map sum) ·map inits · tails

= { since map f ·map g = map (f · g) }
max ·map (max ·map sum · inits) · tails .

Recall the definition scanr f e = map (foldr f e) ·
tails . If we can transform max ·map sum · inits into
a fold, we can turn the algorithm into a scanr , which
has a faster implementation.

Maximum Prefix Sum
Concentrate on max ·map sum · inits :

max ·map sum · inits
= { def. of inits , let ini x xss = [ ] :map (x :) xss }
max ·map sum · foldr ini [[ ]]

= { fold fusion, see below }
max · foldr zplus [0] .

The fold fusion works because:

map sum (ini x xss)

= map sum ([ ] : map (x :) xss)

= 0 : map (sum · (x :)) xss

= 0 : map (x+) (map sum xss) .

Define zplus x yss = 0 :map (x+) yss .

Maximum Prefix Sum, 2nd Fold Fusion
Concentrate on max ·map sum · inits :

max ·map sum · inits
= { def. of inits , let ini x xss = [ ] :map (x :) xss }
max ·map sum · foldr ini [[ ]]

= { fold fusion, zplus x yss = 0 :map (x+) yss }
max · foldr zplus [0]

= { fold fusion, let zmax x y = 0 ‘max ‘ (x + y) }
foldr zmax 0 .

The fold fusion works because ↑ distributes into (+):

max (0 : map (x+) xs)

=0 ↑ max (map (x+) xs)

=0 ↑ (x+max xs) .

Back to Maximum Segment Sum
We reason:

max ·map sum · concat ·map inits · tails
= { since map f · concat = concat ·map (map f ) }
map · concat ·map (map sum) ·
map inits · tails

= { since max · concat = max ·map max }
max ·map max ·map (map sum) ·
map inits · tails

= { since map f ·map g = map (f · g) }
max ·map (max ·map sum · inits) · tails

= { previous reasoning }
max ·map (foldr zmax 0) · tails

= { introducing scanr }
max · scanr zmax 0 .
.

Maximum Segment Sum in Linear Time!

• We have derived mss = max · scanr zmax 0,
where zmax x y = 0 ↑ (x+ y).

• The algorithm runs in linear time, but takes linear
space.

• A tupling transformation eliminates the need for
linear space.

mss = fst ·maxhd · scanr zmax 0

where maxhd xs = (max xs, head xs). We omit
this last step in the lecture.

• The final program is mss = fst · foldr step (0, 0),
where step x (m, y) = ((0 ↑ (x + y)) ↑ m, 0 ↑
(x+ y)).

6 Red-Black Tree
• A self-balancing binary search tree, often used to

represent sets.

• Supports O(log n)-time searching, insertion, and
deletion.

• One possible representation:

data RBTree a = E |
N Color (RBTree a) a (RBTree a) ,

data Color = R | B .

Constraints

• It is a binary search tree.

– In N t x u , every label in t is less than x ,
every label in u is more than x . The same
holds for t and u .

• Each node is either colored red or black.

– E is implicitly considered black.

• The root is black.

• Red nodes do not have red children.

• The number of black nodes from the root to each
leaf is the same.

Searching
Searching in a red-black tree is just like that in a

binary search tree:

search :: Int→ RBTree Int→ Bool
search E = False
search (N t x u) | k < x = ...

| k = = x = ...
| k > x = ...

Exercise: what if we want to return the found ele-
ment in a Maybe?

Insertion

• To insert a new element, perform a search to de-
termine where to insert.

• The inserted node shall have color red.

• This would temporarily break the constraint that
a red node shall not have a red children. We per-
form balancing upwards to restore the constraint.
See the next slide.

• Finally we set the root to black.

Figure 2: Red-black tree balancing [Oka99].

Tree Balancing

• The re-balancing strategy is not unique.

• The strategy we will consider, shown in Figure 2,
was presented by Okasaki [Oka99].

• Having only four rules, it is significantly simpler
than those you’d find in most textbooks (which
needs 8 rules or more)!

• Why?

• More will be discussed in the practicals.

A GHCi Commands

⟨statement⟩ evaluate/run ⟨statement⟩
: repeat last command
:\{\n ..lines.. \n:\}\n} multiline command
:add [*]<module> ... add module(s) to the current target set
:browse[!] [[*]<mod>] display the names defined by module <mod> (!: more details; *: all

top-level names)
:cd <dir> change directory to <dir>

:cmd <expr> run the commands returned by <expr>::IO String

:ctags[!] [<file>] create tags file for Vi (default: "tags") (!: use regex instead of line
number)

:def <cmd> <expr> define command :<cmd> (later defined command has precedence,
::<cmd> is always a builtin command)

:edit <file> edit file
:edit edit last module
:etags [<file>] create tags file for Emacs (default: "TAGS")
:help, :? display this list of commands
:info [<name> ...] display information about the given names
:issafe [<mod>] display safe haskell information of module <mod>
:kind <type> show the kind of <type>
:load [*]<module> ... load module(s) and their dependents
:main [<arguments> ...] run the main function with the given arguments
:module [+/-] [*]<mod> ... set the context for expression evaluation
:quit exit GHCi
:reload reload the current module set
:run function [<arguments> ...] run the function with the given arguments
:script <filename> run the script <filename>
:type <expr> show the type of <expr>
:undef <cmd> undefine user-defined command :<cmd>

:!<command> run the shell command <command>

Commands for debugging

:abandon at a breakpoint, abandon current computation
:back go back in the history (after :trace)
:break [<mod>] <l> [<col>] set a breakpoint at the specified location
:break <name> set a breakpoint on the specified function
:continue resume after a breakpoint
:delete <number> delete the specified breakpoint
:delete * delete all breakpoints
:force <expr> print <expr>, forcing unevaluated parts
:forward go forward in the history (after :back)
:history [<n>] after :trace, show the execution history
:list show the source code around current breakpoint
:list identifier show the source code for <identifier>
:list [<module>] <line> show the source code around line number <line>
:print [<name> ...] prints a value without forcing its computation
:sprint [<name> ...] simplifed version of :print
:step single-step after stopping at a breakpoint
:step <expr> single-step into <expr>

:steplocal single-step within the current top-level binding
:stepmodule single-step restricted to the current module
:trace trace after stopping at a breakpoint

:trace <expr> evaluate <expr> with tracing on (see :history)

Commands for changing settings

:set <option> ... set options
:seti <option> ... set options for interactive evaluation only
:set args <arg> ... set the arguments returned by System.getArgs

:set prog <progname> set the value returned by System.getProgName

:set prompt <prompt> set the prompt used in GHCi
:set editor <cmd> set the command used for :edit
:set stop [<n>] <cmd> set the command to run when a breakpoint is hit
:unset <option> ... unset options

Options for and

+m allow multiline commands
+r revert top-level expressions after each evaluation
+s print timing/memory stats after each evaluation
+t print type after evaluation
-<flags> most GHC command line flags can also be set here (eg. -v2,

-fglasgow-exts, etc). For GHCi-specific flags, see User’s Guide,
Flag reference, Interactive-mode options.

Commands for displaying information

:show bindings show the current bindings made at the prompt
:show breaks show the active breakpoints
:show context show the breakpoint context
:show imports show the current imports
:show modules show the currently loaded modules
:show packages show the currently active package flags
:show language show the currently active language flags
:show <setting> show value of <setting>, which is one of [args, prog, prompt,

editor, stop]
:showi language show language flags for interactive evaluation

